# 设定源文件文件夹
aux_source_directory(lib LIBEEMD_SRC)

# 以下部分为库的编译
# 注意目标名必须唯一 所以不能直接生成相同名称的动态库与静态库
# 注意此处不必为目标名称添加lib前缀和相应后缀，cmake会自行添加
add_library(eemd SHARED ${LIBEEMD_SRC})
# 首先添加静态库的生成命令
add_library(eemd_static STATIC ${LIBEEMD_SRC})
# 设置静态库的输出名称从而获得与动态库名称相同的静态库
set_target_properties(eemd_static PROPERTIES OUTPUT_NAME "eemd")
# 设置输出目标属性以同时输出动态库与静态库
set_target_properties(eemd PROPERTIES CLEAN_DIRECT_OUTPUT 1)
set_target_properties(eemd_static PROPERTIES CLEAN_DIRECT_OUTPUT 1)
# 设置动态库的版本号
set_target_properties(eemd PROPERTIES VERSION ${PROJECT_VERSION} SOVERSION ${PROJECT_VERSION_MAJOR}.${PROJECT_VERSION_MINOR})
# 设置库文件的输出地址
set(LIBRARY_OUTPUT_PATH ${PROJECT_BINARY_DIR}/lib)

# 设置编译选项
set(CMAKE_C_FLAGS "${CMAKE_C_FLAGS} -O3")

# 添加GSL库的编译命令
find_package(GSL REQUIRED)
if(GSL_FOUND)
	message(STATUS "GSL Found.")
	include_directories(${GSL_INCLUDE_DIRS})
	target_link_libraries(eemd PUBLIC GSL::gsl)
	target_link_libraries(eemd_static GSL::gsl)
endif()

# 添加openmp的编译命令 设置编译选项
find_package(OpenMP REQUIRED)
if (OpenMP_C_FOUND)
	message(STATUS "OpenMP Found.")
	set(CMAKE_C_FLAGS "${CMAKE_C_FLAGS} ${OpenMP_C_FLAGS}")
	set(CMAKE_EXE_LINKER_FLAGS "${CMAKE_EXE_LINKER_FLAGS} ${OpenMP_EXE_LINKER_FLAGS}")
	set(CMAKE_SHARED_LINKER_FLAGS "${CMAKE_SHARED_LINKER_FLAGS} ${OpenMP_SHARED_LINKER_FLAGS}")
	target_link_libraries(eemd PUBLIC OpenMP::OpenMP_C)
	target_link_libraries(eemd_static OpenMP::OpenMP_C)
endif()

set(CONFIG_FILE_PATH lib/cmake/${PROJECT_NAME})

configure_package_config_file(${PROJECT_SOURCE_DIR}/${PROJECT_NAME}Config.cmake.in 
	${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
	INSTALL_DESTINATION ${CONFIG_FILE_PATH}
	NO_CHECK_REQUIRED_COMPONENTS_MACRO)

write_basic_package_version_file(${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
	VERSION ${PROJECT_VERSION}
	COMPATIBILITY SameMajorVersion)

# 库的安装命令
if(WIN32)
	install(TARGETS eemd DESTINATION lib)
	install(TARGETS eemd_static DESTINATION lib)
else()
	install(TARGETS eemd eemd_static
		EXPORT ${PROJECT_NAME}Targets
		LIBRARY DESTINATION lib
		ARCHIVE DESTINATION lib)
	install(EXPORT ${PROJECT_NAME}Targets
	    DESTINATION ${CONFIG_FILE_PATH})
	install(FILES
        ${CMAKE_BINARY_DIR}/${PROJECT_NAME}Config.cmake
        ${CMAKE_BINARY_DIR}/${PROJECT_NAME}ConfigVersion.cmake
        DESTINATION ${CONFIG_FILE_PATH})
endif()
# 头文件安装命令
install(FILES lib/eemd.h DESTINATION include)

# 以下部分为例子程序的编译
# 设置可执行文件的输出地址
set(EXECUTABLE_OUTPUT_PATH ${PROJECT_BINARY_DIR}/bin)

# 例子的编译方法
macro(add_sample name)
	# 添加可执行文件 命令行
	add_executable(${name} examples/${name}.c)
	# 为安装文件添加动态库的搜索地址 在Windows下并没有什么用 直接忽略
	set_target_properties(${name} PROPERTIES INSTALL_RPATH ${CMAKE_INSTALL_PREFIX}/lib)
	# 链接动态库
	target_link_libraries(${name} PUBLIC eemd)
endmacro()

macro(add_sample_cxx name)
	# 添加可执行文件 命令行
	add_executable(${name} examples/${name}.cpp)
	# 为安装文件添加动态库的搜索地址 在Windows下并没有什么用 直接忽略
	set_target_properties(${name} PROPERTIES INSTALL_RPATH ${CMAKE_INSTALL_PREFIX}/lib)
	# 链接动态库
	target_link_libraries(${name} PUBLIC eemd)
endmacro()

add_sample(eemd_example)
add_sample_cxx(eemd_example2)
add_sample(ceemdan_example)
